Discrete Heat Transfer

```
> restart;
> with(plots):
```

Wire-Mesh Model

We consider a 2D mesh of nodes connected by thin wires.
In equilibrium, the temperature of a node is equal to the mean of its neighbours.
At boundaries, we assume the temperature to be 0.

N and M denote the number of inner nodes in x- and y-direction:

```
> N := 4; M := 4;
```

$$N := 4$$

$$M := 4 \qquad (1.1)$$

x and y will hold the coordinates of the mesh nodes:

```
> x := array(0..N, [seq(n/N, n=0..N)]):
  y := array(0..M, [seq(m/M, m=0..M)]):
```

And the $(N+1)\,(M+1)$ unknowns $u_{n,\,m}$ are saved here:

```
> unbek := seq(seq(u[n,m], n=0..N), m=0..M);
```

$$unbek := u_{0,\,0}, u_{1,\,0}, u_{2,\,0}, u_{3,\,0}, u_{4,\,0}, u_{0,\,1}, u_{1,\,1}, u_{2,\,1}, u_{3,\,1}, u_{4,\,1}, u_{0,\,2}, u_{1,\,2}, u_{2,\,2}, u_{3,\,2}, u_{4,\,2}, u_{0,\,3}, u_{1,\,3}, u_{2,\,3}, u_{3,\,3}, u_{4,\,3}, u_{0,\,4}, u_{1,\,4}, u_{2,\,4}, u_{3,\,4}, u_{4,\,4} \qquad (1.2)$$

The function f is used to compute the right-hand sides (heat sources/drains):

```
> f := (x,y) -> 0;
```

$$f := (x, y) \rightarrow 0 \qquad (1.3)$$

We set up the respective system of linear equations:

```
> gls := { seq( seq(
              -u[n-1,m]-u[n,m-1] + 4*u[n,m]
              -u[n+1,m]-u[n,m+1] = f(x[n],y[n]),
           n=1..N-1), m=1..M-1),
           seq( u[n,0]=evalf(sin(n*Pi/N)), n=0..N ),
           seq( u[n,M]=0, n=0..N ),
           seq( u[0,m]=0, m=1..M-1),
           seq( u[N,m]=0, m=1..M-1) };
```

$$gls := \{u_{0,\,0} = 0., u_{0,\,1} = 0, u_{0,\,2} = 0, u_{0,\,3} = 0, u_{0,\,4} = 0, u_{1,\,0} = 0.7071067810, u_{1,\,4} = 0, u_{2,\,0} = 1., u_{2,\,4} = 0, u_{3,\,0} = 0.7071067810, u_{3,\,4} = 0, u_{4,\,0} = 0., u_{4,\,1} = 0, u_{4,\,2} = 0, u_{4,\,3} = 0, u_{4,\,4} = 0, -u_{0,\,1} - u_{1,\,0} + 4\,u_{1,\,1} - u_{2,\,1} - u_{1,\,2} = 0, -u_{0,\,2} - u_{1,\,1} + 4\,u_{1,\,2} - u_{2,\,2} - u_{1,\,3} = 0, -u_{0,\,3} - u_{1,\,2} + 4\,u_{1,\,3} - u_{2,\,3} - u_{1,\,4} = 0, -u_{1,\,1} - u_{2,\,0} + 4\,u_{2,\,1} - u_{3,\,1} - u_{2,\,2} = 0, -u_{1,\,2} - u_{2,\,1} + 4\,u_{2,\,2} - u_{3,\,2} - u_{2,\,3} = 0, -u_{1,\,3} - u_{2,\,2} + 4\,u_{2,\,3} - u_{3,\,3} - u_{2,\,4} = 0, -u_{2,\,1} - u_{3,\,0} + 4\,u_{3,\,1} - u_{4,\,1} - u_{3,\,2} = 0, -u_{2,\,2} - u_{3,\,1} + 4\,u_{3,\,2} - u_{4,\,2} - u_{3,\,3} = 0, -u_{2,\,3} - u_{3,\,2} + 4\,u_{3,\,3} - u_{4,\,3} - u_{3,\,4} = 0\} \qquad (1.4)$$

... and use Maple to solve the system:

```
> lsg := solve(gls,{unbek});
```

$$lsg := \{u_{0,0} = 0., u_{0,1} = 0., u_{0,2} = 0., u_{0,3} = 0., u_{0,4} = 0., u_{1,0} = 0.7071067810, u_{1,1} = 0.3318120616, u_{1,2} = 0.1508883476, u_{1,3} = 0.05835298129, u_{1,4} = 0., u_{2,0} = 1., u_{2,1} = 0.4692531177, u_{2,2} = 0.2133883476, u_{2,3} = 0.08252357755, u_{2,4} = 0., u_{3,0} = 0.7071067810, u_{3,1} = 0.3318120616, u_{3,2} = 0.1508883476, u_{3,3} = 0.05835298129, u_{3,4} = 0., u_{4,0} = 0., u_{4,1} = 0., u_{4,2} = 0., u_{4,3} = 0., u_{4,4} = 0.\} \quad (1.5)$$

Plotting the solution

We use a helper function to compute that returns a triple of x- and y-coordinate and value of the function at this point:

```
> punkt := (i,j) -> [x[i],y[j],subs(lsg, u[i,j])];
```

$$punkt := (i, j) \rightarrow [x_i, y_j, subs(lsg, u_{i,j})] \quad (1.1.1)$$

For the center of the mesh, we get:

```
> punkt(N/2, M/2);
```

$$\left[\frac{1}{2}, \frac{1}{2}, 0.2133883476\right] \quad (1.1.2)$$

This is a plot of all solution points:

```
> punkte := [seq( seq(punkt(i,j), i=0..N),
                 j=0..N)];
> pointplot3d(punkte, symbol=CIRCLE, axes=BOXED);
```

$$punkte := \left[[0, 0, 0.], \left[\frac{1}{4}, 0, 0.7071067810\right], \left[\frac{1}{2}, 0, 1.\right], \left[\frac{3}{4}, 0, 0.7071067810\right], [1, 0, 0.], \left[0, \frac{1}{4}, 0.\right], \left[\frac{1}{4}, \frac{1}{4}, 0.3318120616\right], \left[\frac{1}{2}, \frac{1}{4}, 0.4692531177\right], \left[\frac{3}{4}, \frac{1}{4}, 0.3318120616\right], \left[1, \frac{1}{4}, 0.\right], \left[0, \frac{1}{2}, 0.\right], \left[\frac{1}{4}, \frac{1}{2}, 0.1508883476\right], \left[\frac{1}{2}, \frac{1}{2}, 0.2133883476\right], \left[\frac{3}{4}, \frac{1}{2}, 0.1508883476\right], \left[1, \frac{1}{2}, 0.\right], \left[0, \frac{3}{4}, 0.\right], \left[\frac{1}{4}, \frac{3}{4}, 0.05835298129\right], \left[\frac{1}{2}, \frac{3}{4}, 0.08252357755\right], \left[\frac{3}{4}, \frac{3}{4}, 0.05835298129\right], \left[1, \frac{3}{4}, 0.\right], [0, 1, 0.], \left[\frac{1}{4}, 1, 0.\right], \left[\frac{1}{2}, 1, 0.\right], \left[\frac{3}{4}, 1, 0.\right], [1, 1, 0.]\right]$$



But it's more convenient to plot a 2D linear interpolation of the solution:

```
> surfdata([seq([seq(punkt(i,j), i=0..N)],
                 j=0..N)],axes=BOXED);
```



Finite Volume Model

Set up the unknowns

```
> N := 4:  M := 4:
> x := array(0..N, [seq(n/N, n=0..N)]):
  y := array(0..M, [seq(m/M, m=0..M)]):
> unbek := seq(seq(u[n,m], n=0..N), m=0..M);
```

$$unbek := u_{0,0}, u_{1,0}, u_{2,0}, u_{3,0}, u_{4,0}, u_{0,1}, u_{1,1}, u_{2,1}, u_{3,1}, u_{4,1}, u_{0,2}, u_{1,2}, u_{2,2}, u_{3,2}, u_{4,2}, u_{0,3}, u_{1,3}, u_{2,3}, u_{3,3}, u_{4,3}, u_{0,4}, u_{1,4}, u_{2,4}, u_{3,4}, u_{4,4} \quad \textbf{(2.1.1)}$$

Right-hand side:

```
> f := (x,y) -> 0;
```

$$f := (x, y) \rightarrow 0 \quad \textbf{(2.1.2)}$$

System of linear equations:

```
> part1 := { seq( seq(
              -u[n-1,m]-u[n,m-1] + 4*u[n,m]
              -u[n+1,m]-u[n,m+1] = f(x[n],y[n]),
           n=1..N-1), m=1..M-1),
           seq( u[n,0]=evalf(sin(n*Pi/N)), n=0..M )};
```

$$part1 := \{u_{0,0} = 0., u_{1,0} = 0.7071067810, u_{2,0} = 1., u_{3,0} = 0.7071067810, u_{4,0} = 0., -u_{0,1} - u_{1,0} + 4\,u_{1,1} - u_{2,1} - u_{1,2} = 0, -u_{0,2} - u_{1,1} + 4\,u_{1,2} - u_{2,2} - u_{1,3} = 0, -u_{0,3} - u_{1,2} + 4\,u_{1,3} - u_{2,3} - u_{1,4} = 0, -u_{1,1} - u_{2,0} + 4\,u_{2,1} - u_{3,1} - u_{2,2} = 0, -u_{1,2} - u_{2,1} + 4\,u_{2,2} - u_{3,2} - u_{2,3} = 0, -u_{1,3} - u_{2,2} + 4\,u_{2,3} - u_{3,3} - u_{2,4} = 0, -u_{2,1} - u_{3,0} + 4\,u_{3,1} - u_{4,1} - u_{3,2} = 0, -u_{2,2} - u_{3,1} + 4\,u_{3,2} - u_{4,2} - u_{3,3} = 0, -u_{2,3} - u_{3,2} + 4\,u_{3,3} - u_{4,3} - u_{3,4} = 0\} \quad \textbf{(2.1.3)}$$

```
> neumpart := {
           seq( -u[n-1,M]-u[n,M-1] + 3*u[n,M]
                -u[n+1,M] = 0, n=1..N-1 ),
           seq( -u[0,m-1] + 3*u[0,m]
              -u[1,m]-u[0,m+1] = 0, m=1..M-1),
           seq( -u[N,m-1] + 3*u[N,m]
                -u[N-1,m]-u[N,m+1] = 0, m=1..M-1),
           -u[N,M-1] + 2*u[N,M]-u[N-1,M] = 0,
           -u[0,M-1] + 2*u[0,M]-u[1,M] = 0 };
```

$$neumpart := \{-u_{0,3} + 2\,u_{0,4} - u_{1,4} = 0, -u_{4,3} + 2\,u_{4,4} - u_{3,4} = 0, -u_{0,0} + 3\,u_{0,1} - u_{1,1} - u_{0,2} = 0, -u_{0,1} + 3\,u_{0,2} - u_{1,2} - u_{0,3} = 0, -u_{0,2} + 3\,u_{0,3} - u_{1,3} - u_{0,4} = 0, -u_{0,4} - u_{1,3} + 3\,u_{1,4} - u_{2,4} = 0, -u_{1,4} - u_{2,3} + 3\,u_{2,4} - u_{3,4} = 0, -u_{2,4} - u_{3,3} + 3\,u_{3,4} - u_{4,4} = 0, -u_{4,0} + 3\,u_{4,1} - u_{3,1} - u_{4,2} = 0, -u_{4,1} + 3\,u_{4,2} - u_{3,2} - u_{4,3} = 0, -u_{4,2} + 3\,u_{4,3} - u_{3,3} - u_{4,4} = 0\} \quad \textbf{(2.1.4)}$$

```
> gls := part1 union neumpart:
```

and solve the system:

```
> lsg := solve(gls,{unbek});
```

$$lsg := \{u_{0,0} = 0., u_{0,1} = 0.3272704168, u_{0,2} = 0.4322145534, u_{0,3} = 0.4658429871, \quad \textbf{(2.1.5)}$$

$u_{0,4} = 0.4757285755$, $u_{1,0} = 0.7071067810$, $u_{1,1} = 0.5495966971$, $u_{1,2} = 0.5035302564$, $u_{1,3} = 0.4895858325$, $u_{1,4} = 0.4856141638$, $u_{2,0} = 1.$, $u_{2,1} = 0.6604793341$, $u_{2,2} = 0.5427239424$, $u_{2,3} = 0.5033559227$, $u_{2,4} = 0.4915280834$, $u_{3,0} = 0.7071067810$, $u_{3,1} = 0.5495966971$, $u_{3,2} = 0.5035302564$, $u_{3,3} = 0.4895858325$, $u_{3,4} = 0.4856141638$, $u_{4,0} = 0.$, $u_{4,1} = 0.3272704168$, $u_{4,2} = 0.4322145534$, $u_{4,3} = 0.4658429871$, $u_{4,4} = 0.4757285755\}$

```
> punkt := (i,j) -> [x[i],y[j],subs(lsg, u[i,j])]:
> punkte := [seq( seq(punkt(i,j), i=0..N),
                 j=0..N)]:
  surfdata([seq([seq(punkt(i,j), i=0..N)],
                j=0..N)],axes=BOXED);
```